/**
 * Copyright (c) 2009 Teoti Graphix, LLC See:
 *     http://tkarchitecture.teotigraphix.org
 * 
 * Permission is hereby granted, free of charge, to any person
 * obtaining a copy of this software and associated documentation
 * files (the "Software"), to deal in the Software without
 * restriction, including without limitation the rights to use,
 * copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the
 * Software is furnished to do so, subject to the following
 * conditions:
 * 
 * The above copyright notice and this permission notice shall be
 * included in all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 * EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 * OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 * NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 * HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 * WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 * FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 * OTHER DEALINGS IN THE SOFTWARE.
 * 
 * @author Michael Schmalle mschmalle@teotigraphix.com
 */
package com.example.acme.model.proxy
{

import com.example.acme.ApplicationFacade;
import com.example.acme.model.business.UserLoginDelegate;
import com.example.acme.model.vo.User;

import mx.rpc.Responder;

import org.puremvc.as3.multicore.patterns.proxy.Proxy;

/**
 * Activates the login service and listens for either result of failure.
 * 
 * @author Michael Schmalle
 * @copyright Teoti Graphix, LLC
 */
public class UserProxy extends Proxy
{
	//--------------------------------------------------------------------------
	//
	//  Public Class :: Properties
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */	
	public static const NAME:String = "UserProxy";
	
	//--------------------------------------------------------------------------
	//
	//  Private :: Variables
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */
	private var delegate:UserLoginDelegate;
	
	/**
	 * @private
	 */
	private var pendingUser:User;
	
	//--------------------------------------------------------------------------
	//
	//  Constructor
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */
	public function UserProxy()
	{
		super( NAME );
	}
	
	//--------------------------------------------------------------------------
	//
	//  Overridden Public :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Creates the single <code>UserLoginDelegate</code> instance.
	 */
	override public function onRegister():void
	{
		delegate = new UserLoginDelegate();
	}
	
	//--------------------------------------------------------------------------
	//
	//  Public API :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Logs in the user using the <code>UserLoginDelegate</code> instance.
	 * 
	 * @param user A User instance with username and password for use with the
	 * delegate's service API.
	 */
	public function loginUser( user:User ):void
	{
		// don't let logins happen if there is a pending login
		if (pendingUser)
			return;
		
		// break reference from the original caller
		pendingUser = user.clone();
		
		// call the service, pass the responder pointing to the 
		// result and fault handlers
		delegate.loginUser( pendingUser, new Responder( loginResult, loginFault ) );
	}
	
	/**
	 * Creates a temp User instance for use with <code>loginUser()</code>.
	 * 
	 * @param username A string indicating the user's name credentials.
	 * @param password A string indicating the user's password credentials.
	 * 
	 * @return A temp User instance.
	 */
	public function createTempUser( username:String, password:String ):User
	{
		return new User( username, password );
	}
	
	//--------------------------------------------------------------------------
	//
	//  Private :: Handlers
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */
	private function loginResult( result:User ):void
	{
		// stop the pending session
		pendingUser = null;
		// set our proxy User data
		setData( result );
		// send notification to the observers
		sendNotification( ApplicationFacade.USER_LOGIN_COMPLETE, result );
	}
	
	/**
	 * @private
	 */
	private function loginFault( fault:Object ):void
	{
		// impl, if fault notify
	}
}
}